

-SINCLAIR INFORMATION NOUVELLES-

No.4 novembre 85 \$2.00

Rédacteur: Réal Gagnon (544 Hermine #2 Québec P.Q. G1N 2G6)

Ce bulletin est conçu avec l'aide d'un traitement de texte

TASWORD II de Tasman Software (version 2068)

et d'une imprimante SMITH-CORONA, Fastext 80.

** DANS CE NUMERO **

- *PROGRAMME :LA VOIX DU 2068, faites parler votre 2068
- *PROGRAMME :CONVERSION HEX->DEC & DEC->HEX
- *PROGRAMME :TITRE 3D, créer des pages-titres en 3D.
+ ECRAN 3D, " " " " " "
- *PROGRAMME :SURFACE PLANE, un peu de graphisme
+ BELLE CASSETTE, embellissez vos cassettes
- *TECHNIQUE :Ecrire à l'écran en langage machine
- *LOGICIEL :ANT ATTACK (Spectrum)
CHEQUERED FLAG (T52068)
MANIC MINER (Spectrum)

*PLUS des trucs pour le 2068 et l'actualité.

SINCLAIR NEWS

LANGAGES DISPONIBLES POUR LE TS2068

Pour ceux que le BASIC n'intéresse plus HAWG SOFTWARE propose 2 langages de programmation pour le 2068 non-modifié, il s'agit d'un FORTH et d'un PASCAL. En ce qui concerne le FORTH, il s'agit évidemment d'un Fig-FORTH et il vient avec un manuel de 41 pages, son prix est environ 30\$ US. Le PASCAL est une adaptation du HiSoft PASCAL qui existe pour le SPECTRUM. Il s'agit ici d'un PASCAL très complet et il vient avec un manuel de plus de 80 pages. Pour plus d'informations au sujet du FORTH et du PASCAL, écrire à HAWG SOFTWARE, P.O. Drawer 7680, Little Rock, Arkansas 72217, USA. Le FORTH est également disponible chez: THE FORTH SOURCE, Mountain View Press inc., Mountain View, CA 94040, USA.

INFORMATIQUE-TELEVISION

Pour ceux qui possèdent un cabloselecteur, le dimanche à 18h00, TVFQ chaîne 99 (canal 10) présente une émission qui s'appelle PIC & POKE & COLEGRAM. Cette émission traite de la micro-informatique en France.

Plus près de nous au canal communautaire (canal 9), M. Tony Caron effectue un retour avec son émission JEUX VIDEO. Quoique que consacrée presque exclusivement aux logiciels pour COMMODORE ou ATARI, cette émission est tout même très intéressante.

INFORMATIQUE-BIBLIOTHEQUE

La Bibliothèque Centrale possède maintenant une nouvelle section pour les amateurs de micro: la LOGITHEQUE. Cette section contient une collection de livres informatiques assez intéressante. A remarquer les versions françaises des excellents livres de Rodney ZAKS sur le Z80. J'ai remarqué également quelques bouquins sur le ZX81 et le SPECTRUM. Toutes les revues informatiques sont maintenant à la LOGITHEQUE. Il y a également quelques ordinateurs (des PCs...) sur lesquels il est possible de s'initier à des logiciels dits "professionnels" (genre dBase II).

TS2068: RESSURECTION?

Il y a une grosse rumeur qui circule en ce moment en Angleterre voulant que Sinclair soit sur le point de mettre sur le marché un SPECTRUM de 128K. Dans POPULAR COMPUTING WEEKLY (Vol. 4, No. 32), il y a une description de ce SPECTRUM 128K, si on se base sur cette description ce nouveau SPECTRUM est bel et bien un TS2068. Le 128K est le résultat d'un système de BANK SWITCHING, système aussi présent sur le TS2068. Un chip de son (AY38912), 3 modes graphiques et 2 ports JOYSTICKS complètent le "nouveau" SPECTRUM. Une petite ressemblance avec le 2068, n'est-ce pas !?

PROGRAMME: HEX-→DEC DEC-→HEX

Deux petits utilitaires bien utiles qui convertissent n'importe quel nombre hexadecimal en son équivalent decimal ou vice et versa.

```
1 REM HEX-→DEC
20 LET DEC=0: LET A=10: LET B=11: LET C=12: LET D=13: LET E=14
: LET F=15
30 INPUT "HEX CODE: "; LINE H$
35 FOR I=1 TO LEN H$-1
40 LET DEC=DEC+ VAL H$(I)*16^ LEN H$-I: NEXT I
50 LET DEC=DEC+ VAL H$(I)
60 PRINT H$,DEC: GO TO 20

1 REM DEC-→HEX
10 INPUT "DEC CODE:";a: PRINT a,
20 GO SUB 100: PRINT a$: GO TO 10
100 LET a$=""
110 LET x= INT (a/16): LET y=a-x*16: LET a=x
113 LET y=y+48+(7 AND y>9): LET a$= CHR$ y+a$
170 IF a THEN GO TO 110
180 IF INT ( LEN a$/2) <> LEN a$/2 THEN LET a$="0"+a$
190 RETURN
```

Programme: Belle Cassette

Ce programme vous permet de créer sur votre imprimante des "pages couvertures" pour vos cassette.

```
1 REM      PRESENTATION-CASSETTE
1000 LET l$="_____ "
1200 DIM b$(14,32): LET a$="          couper ici "
5005 PRINT AT 21,0;"Le nom du programme? (Max. 16)"
5010 INPUT AT 0,0; LINE P$' USR 2361' USR 2361
5020 PRINT AT 21,0;"Le type d'ordinateur? (Max. 10)"
5025 INPUT AT 0,0; LINE O$' USR 2361' USR 2361
5030 FOR i=1 TO 13: PRINT AT 21,0;"Le texte #";i,"(Max. 32)": I
NPUT AT 0,0; LINE T$' USR 2361' USR 2361
5035 IF LEN T$ THEN LET b$(i,(16-LEN T$/2)+1 TO )=t$
5040 NEXT I: LPRINT a$
6020 CLS : PLOT 0,175: DRAW 255,0: LET FLAG=1: LET xx1=(256-(LE
N P$*16))/2-8: LET yy=170: LET y$=P$: GO SUB 9000
6025 PLOT 0,YY-2: DRAW 255,0: PRINT AT 6,0;
6030 FOR i=1 TO 13: PRINT b$(i): NEXT i
6040 LET FLAG=0: LET xx1=(256-(LEN O$*8))/2-8: LET yy=16: LET y
$=O$: GO SUB 9000
6500 PRINT #0;1$: COPY
6520 CLS : LET Y$=p$+" (" +0$+*)": LET XX1=(256-(LEN Y$*8))/2:
LET yy=175: CLS : GO SUB 9000: PLOT 0,yy-2: DRAW 255,0
6600 PRINT AT 6,0;c$,"CASSETTE""1$: COPY : STOP
9005 LET xx2=xx1: PRINT AT 21,0;y$
9010 FOR y=7 TO 0 STEP -1: FOR x=0 TO 255
9020 IF POINT (x,y) THEN GO SUB 9500: IF FLAG THEN LET YY2=YY
: LET YY=YY-2: GO SUB 9500: LET YY=YY2
9030 LET xx2=xx2+1+FLAG: NEXT x: LET yy=yy-2-(2*FLAG): LET xx2=x
x1: NEXT y: PRINT AT 21,0;b$(14, TO LEN y$): RETURN
9500 PLOT xx2,yy: PLOT xx2+1,yy
9501 PLOT xx2,yy+1: PLOT xx2+1,yy+1: RETURN
9999 SAVE "pres_cass": VERIFY "": REM 6A6-o85
```

ESSAI de LOGICIELS

NOM : ANT ATTACK (Spectrum)

TYPE : Jeu

EDITEUR: QUICKSYLVA

Vous etes dans une ville appelée ANTESCHER et vous devez sauvez une jeune fille (ou un jeune garçon selon le cas),mais attention il y a des fourmis géantes qui vous poursuivent sans relache. ANT ATTACK est un classique. L'animation est sublime. Nous avons un vue en 3D (genre ZAXXON),avec possibilité de 4 angles de vue différents. L'action est très rapide. Soyez assuré que ce n'est pas le genre de jeu auquel l'on joue 5 minutes et qu'on veut plus rien savoir par la suite,au contraire ANT ATTACK est passionnant. Les Anglais appellent ce genre de jeu : an addictive game , et ils ont raison ...

NOM : CHEQUERED FLAG (TS2068)

TYPE : Simulation

EDITEUR: PSION

Si votre reve est de conduire une FORMULE 1,et bien CHEQUERED FLAG est pour vous. Vous avez le choix entre 3 types de FORMULE 1 : 2 à transmission manuelle et l'autre (pour les débutants...) à transmission automatique. Vous avez également le choix entre 10 parcours différents. Le but est de compléter le parcours sans faire trop d'accidents et avec le meilleur temps possible. Sur la piste,il faut éviter les éternelles taches d'huile et des éclats de verre brisé. Il faut également surveiller le niveau d'essence et éviter le surchauffement du moteur,si nous avons un ennui mécanique,il est possible d'arreter à un puit de ravitaillement. L'animation et les effets sonores sont très bons. Un point très important pour terminer:cette version de CHEQUERED FLAG fonctionne avec un TS2068 non-modifié,donc on peut enfin utiliser un JOYSTICK .

NOM : MANIC MINER (Spectrum)

TYPE : Jeu

EDITEUR: SOFTWARE PROJECT

Un autre classique du SPECTRUM. Il s'agit de guider WILLY à

travers différentes cavernes, dans chaque caverne WILLY doit trouver un certain nombre de clés pour pouvoir passer à la caverne suivante. Ce jeu demande de la stratégie et de l'adresse ... et beaucoup de patience. Il y a une dizaine de tableaux différents. L'animation est d'une qualité incroyable et l'action est très rapide. Donc plusieurs heures de bonheur assuré avec MANIC MINER pour Spectrum...

Technique: ECRIRE en C.M.

L'assembleur du Z 80 possède une instruction spéciale pour écrire du texte à l'écran. Cette instruction a pour nom RST 16, RST signifie ReSTart. L'utilisation de cette instruction est très simple. En premier lieu, il suffit de charger l'Accumulateur avec le code ASCII (qui correspond au caractère que l'on veut afficher) et on fait RST 16. Si on veut afficher la lettre "A", dont le code ASCII est 65, nous faisons: LD A,65 et RST 16.

Mais que faire si nous voulons afficher un "A" sur la dixième ligne et à la dixième colonne. Regardons dans le 2068 USER MANUAL à la page 239 la liste des codes ASCII. Nous remarquons que le code 22 correspond au caractère de contrôle AT. Ainsi pour afficher sur la ligne 10 et à la colonne 10, nous faisons:

```
LD A,22    ;contrôle AT
RST 16     ;
LD A,10    ;ligne 10
RST 16
LD A,10    ;colonne 10
RST 16
LD A,65    ;code "A"
RST 16
RET
```

Ainsi pour une couleur PAPER différente:

```

LD A,17    ;controle INK
RST 16
LD A,5      ;couleur CYAN
RST 16
LD A,65     ;code "A"
RST 16
RET

```

Comme nous pouvons le constater il serait fastidieux d'afficher un texte complet de cette façon. La solution est d'utiliser une boucle. La ROM du 2068 contient une routine, appelée PR-STRING, qui va nous donné un coup de main pour afficher un long texte. Cette routine est située à l'adresse 8667. Avant d'appeler cette routine nous chargeons BC avec la longueur du texte et DE avec l'adresse de départ des codes ASCII.

```

ORG 40000
LD BC,6     ;6 codes à afficher
LD DE,50000 ;adr. des codes
CALL 8667   ;PR_STRING
RET         ;retour->BASIC
ORG 50000
DEFB 22,20,15 ;AT 20,15
DEFB "S","I","N ;notre message

```

Voici une petite routine qui démontre une technique d'affichage de petit message (ex. messages d'erreur). Nous chargeons A avec le code d'identification du message à afficher. Puis HL avec l'adresse de départ de la zone de recherche, BC est la longueur de cette zone. L'instruction CPIR signifie ComPare Increment & Repeat, elle compare le contenu de HL avec A, si la comparaison n'est pas vrai (flag Z=0) alors HL=HL+1 et BC=BC-1 et on répète.

```

ORG 40000
LD A,1      ;code du message
LD HL,45000 ;adr. de départ
LD BC,6000  ;long. de la zone
CPIR
JP Z,ok     ;A=(HL)-->ok
RET         ;BC=0 -->BASIC

loop INC HL
LD A,(HL)  ;ASCII
RST 16     ;affiche
DJNZ loop  ;BC=0?
RET        ;->BASIC
ORG 50000
mess DEFB 1,3,"S","I","N

```

↓
 ok LD B,(HL) ;B=long. du message
 PUSH BC ;initialisation de
 PUSH HL ;l'écran.
 LD A,2 ;A=1 bas écran
 CALL \$1230* ;*A=2 haut écran*
 POP HL ;A=3 imprimante
 POP BC ;->obligatoire

↓
 DEFB 2,8,"M","A","G
 DEFB "A","Z","I","N
 DEFB "E
 DEFB 3,6,"T","S","2
 DEFB "O","6","8

* = en hexadécimal

Faites POKE 40001,no. code et RAND USR 40000.

NOTE: Pour ma part, la programmation en langage machine ne peut se faire EFFICACEMENT sans un ASSEMBLEUR. C'est pour cette raison que je n'ai pas inclus les équivalents décimaux des mnémoniques.



N'hésitez pas à faire connaître
 vos commentaires sur SIN. Vous
 pouvez laisser un message chez
 GAGNON ELECTRONIQUE si vous le
 désirez. Egalement si vous vou-
 lez écrire un article pour SIN
 soyez assuré que votre geste
 sera GRANDEMENT apprécié.

Programme : LA VOIX DU TS2068

Ce programme, assez surprenant, vous permet de faire parler votre TS2068 et ce, sans l'aide d'un périphérique quelconque. La marche à suivre est très simple. Vous enregistrez (avec votre propre voix) un mot sur cassette. Puis le TS2068 écoute votre voix et la digitalise dans sa mémoire, i.e. la transforme en "1" et en "0". C'est la routine machine "ENTENDRE" qui exécute cette tâche. Une fois mémorisés par le TS2068, la routine machine "PARLER" peut les reproduire à volonté.

Premièrement, tapez le programme BASIC. Ce programme vous permettra de créer facilement vos mots. N'oubliez pas de sauvegarder le programme sur cassette. Ensuite il y a 3 routines machines: CARACTERES AGRANDIS, PARLER et ENTENDRE. Tapez le HEXLOADER, puis les DATAs correspondants à chaque routine. Sur une cassette différente, sauvegardez la routine des caractères agrandis, puis "ENTENDRE" et "PARLER" ensemble: SAVE "CAR_AGR." CODE 32200,300:
SAVE "ENT&PAR" CODE 64900,180:
VERIFY "" CODE: VERIFY "" CODE

DESCRIPTION DES ROUTINES

-CARACTERES AGRANDIS , cette routine permet d'afficher des caractères agrandis. On peut agrandir 2 fois, 3 fois et plus et ce, en hauteur, en largeur ou les deux à la fois. Premièrement, nous pokons le message à agrandir à partir de l'adresse 23320, on termine avec la valeur "255" pour indiquer à la routine la fin du message à agrandir. Cinq autres paramètres sont nécessaires : adresse 23303=départ x
23304=départ y
23305=hauteur
23306=largeur
23307=espace entre 2 caractères

Etudier le programme BASIC aux lignes 4000 à 4030 pour bien saisir le fonctionnement de cette routine. Faites POKE 32314,32 pour INVERSE 0 et POKE 32314,40 pour INVERSE 1. Cette routine est tirée du livre LA PRATIQUE DU ZX SPECTRUM, de Marcel Henrot, édité chez les Editions du P.S.I. .

-ENTENDRE et PARLER sont tirés du livre SUPERCHARGE YOUR

SPECTRUM, de David Webb, édité par Melbourne House. Pour utiliser vos mots dans vos propres programmes BASIC vous avez besoin seulement de la routine "PARLER".

MODE D'EMPLOI

Charger le programme BASIC ainsi que les routines machines. Nous allons enregistrer 3 mots: "TIMEX", "SINCLAIR" et "2068".

Avec votre magnétophone et une cassette, enregistrez les trois mots en notant (si le magnéto a un compteur) le début et la fin de chaque mot. Disons que "TIMEX" débute à 0 et termine à 8, "SINCLAIR" 10 à 20 et "2068" 25 à 30. La première question que le programme vous demande est "Nombre de mots maximums", tapez 3 car nous avons 3 mots sur notre cassette. Puis pour la longueur maximum des mots, tapez 8 car SINCLAIR est le mot le plus long avec ses 8 lettres. Notre premier mot sur cassette est TIMEX. Maintenant le programme nous demande le délai. Ceci signifie la vitesse à laquelle le 2068 va "entendre notre voix". Ceci implique que plus le délai est petit, plus notre voix sera claire, mais entraîne une perte de mémoire plus rapide en résulte malheureusement. Un délai de 20 est généralement acceptable. Vous démarrez votre cassette et appuyez ENTER et le 2068 "écoute". Lorsque le compteur du magnéto arrive à 8 (par exemple) notre mot est terminé. On appuie sur la barre d'espace pour indiquer au 2068 d'arrêter d'écouter. Et on procède de la même façon pour tous les autres mots.

Lorsque nos 3 mots sont enregistrés, un menu de toutes les options possibles apparait. L'option 2 nous fait entendre les mots enregistrés ainsi que les adresses de départ et de fin.

UTILISATION FUTURE DES MOTS

Supposons que vous avez un programme de mastermind et que vous voulez que le 2068 vous dise "MASTERMIND" ou "Une autre partie". Vous n'avez qu'à créer les mots nécessaires et les incorporer au programme MASTERMIND et utiliser la routine "PARLER" pour que les mots "prennent vies".

Exemple: Si le mot "MASTERMIND" débute à l'adresse 60000 et se termine à 58500. La routine "PARLER" se trouvant à 64900.

```

5 CLS
10 REM =====BARRAGE=====
20 REM
30 REM =====VOUS GAGNEZ=====
100 GO SUB 700
105 REM =====INITIALISATION/DIFFICULTE=====
110 PRINT "TAB 9;" "B A R R A G E"
"TAB 8;" "=====
115 BORDER 6
120 PRINT "TAB 5;" "Vous etes en
possession de""TAB 3;" "< 100 bom
bes et 5 avions >" "TAB 2;" "pour
completer votre mission"
125 PRINT "TAB 5;" "Cle 'z' pour
descendre""TAB 5;" "Cle 'm' pour r
emonter"
130 PRINT "" ENTREZ LE NIVEAU
DE DIFFICULTE ""TAB 6;" "( 1-5 )
5 est lent "
150 INPUT sp: IF sp<1 OR sp>5 T
HEN GO TO 150
160 PRINT "TAB 1;" " LE NOMBRE DE
BARRAGE BALLON""TAB 6;" "( 20-60
) 20 est facile"
170 INPUT ba: IF ba<20 OR ba>60
THEN GO TO 170
180 LET cr=0: LET sc=0
190 CLS
200 REM =====INITIALISATION BALLON=====
210 FOR n=1 TO ba
220 LET v=INT (RND*18)+3
230 LET h=INT (RND*32): IF h<3
THEN LET h=3
240 PRINT AT v,h; INK 4;CHR$ 14
8
250 NEXT n
260 PRINT AT 21,0;"BOMBES=" ;sc
: BEEP .2,20
270 IF sc=100 THEN GO TO 500
280 PRINT AT 21,16;" AVION=" ;c
r
290 IF cr=5 THEN GO TO 530
295 PAUSE 100
300 REM =====AVION=====
310 LET v=INT (RND*18)+3
320 FOR h=0 TO 31
330 LET c$=INKEY$
340 IF c$="" THEN PRINT AT v,h-
1;" "
350 IF STICK (1,1)=2 OR c$="z"
THEN PRINT AT v,h-1;" " : LET v=v
+1: IF v>20 THEN LET v=0
360 IF STICK (1,1)=1 OR c$="m"
THEN PRINT AT v,h-1;" " : LET v=v
-1: IF v<0 THEN LET v=0
370 IF h=31 THEN GO TO 600
380 IF ATTR (v,h)=60 THEN PRINT
AT v,h; INK 2; FLASH 1; BRIGHT
1;CHR$ 147: FOR n=1 TO 25: BEEP
.05,-15: BEEP .1,15: NEXT n: PRI
NT AT v,h;" " : LET cr=cr+1: GO T
O 280
390 PRINT AT v,h;CHR$ 144
400 PAUSE sp
410 NEXT h
500 REM =====FIN DE LA PARTIE=====
510 PRINT AT 10,10; INK 3; FLAS
H 1;" MISSION TERMINEE "
520 STOP
530 PRINT AT 10,10; INK 3; FLAS
H 1;" AVION DETRUIT "
540 STOP
600 REM =====SON=====BOMBES=====
610 FOR n=50 TO 29 STEP -.5
620 BEEP .05,n
630 NEXT n
640 FOR x=1 TO 15
650 BEEP .05,-25: BEEP .05,-35:
BEEP .05,-60
660 NEXT x
670 LET sc=sc+10
680 GO TO 260
700 REM =====GRAPHIQUES=====
710 DATA "a",16,24,156,255,156,
24,16,0,"d",145,0,36,0,145,0,36,
145,"e",0,0,120,253,255,253,120,
0
720 FOR n=1 TO 3
730 READ a$
740 FOR a=0 TO 7
750 READ b: POKE USR a$+a,b
760 NEXT a
770 NEXT n
800 RETURN

```

Programme: ECRAN 3D

Ce programme vous permet de créer des pages avec un petit effet en 3D...

```

0> REM ECRAN 3D sur une idee de J. HUNTON _____ GAG-o85
10 DIM b$(1,9): PRINT AT 20,0: INVERSE 1;"3D WORDS base sur u
ne idee de JOHN HUNTON": RANDOMIZE USR 2361: RANDOMIZE USR 2
361
30 INPUT AT 0,0;"Grosueur des lettres (1-3):";grosueur' USR 2
361' USR 2361: IF grosueur<1 OR grosueur>3 THEN GO TO 30:
40 PRINT AT 21,0;"Votre phrase peut contenir '"('9" AND gross
eur=1)+('7" AND grosueur=2)+('6" AND grosueur=3);" au maximum:"
INPUT AT 0,0: LINE a$' USR 2361' USR 2361
50 INPUT AT 0,0;"A partir de quelle ligne""du haut (0-17):";
p' USR 2361' USR 2361: LET p=p*8
55 INPUT AT 0,0;"Encre (0-7):";e'"Papier (0-7):";pap'"' USR 2
361' USR 2361: INK e: PAPER pap: BORDER pap: PAUSE 50: CLS
100 PRINT INK pap; AT 21,0;a$: LET a= LEN a$
110 FOR f=0 TO 8*a-1: FOR n=0 TO 7
120 IF NOT POINT (f,n) THEN GO TO 170
130 PLOT f*(2+grosueur+(.5 AND grosueur=1)),n*4+(grosueur=3)+14
0-(5 AND grosueur=2)-p
145 DRAW 0,-3: DRAW 3,0: DRAW 0,2
150 DRAW -2,0: DRAW 0,-1: DRAW 2,0: DRAW -2,-2
155 DRAW 5,5: DRAW 0,4: DRAW 0,-4: DRAW 4,0: DRAW 0,4: DRAW 0,-
4: DRAW -5,-5
160 DRAW 0,4: DRAW 5,5: DRAW -4,0: DRAW -5,-5
170 NEXT n: NEXT f: PRINT AT 21,0;b$(1)
210 PRINT #0;"[1] COPY [2] STOP [3] ECRITURE"
220 LET i$= INKEY$: IF i$="" THEN GO TO 220
230 IF i$="1" THEN INPUT "": COPY : GO TO 210
235 IF i$="2" THEN STOP
240 IF i$="3" THEN GO TO 1000
250 GO TO 220
1000 INPUT "Grosueur (1-3):";grosueur
1010 PRINT #0;"Longueur max. de '"('9" AND grosueur=1)+('7" AND
grosueur=2)+('6" AND grosueur=3);" lettres.": PAUSE 50
1020 INPUT "Phrase:"; LINE a$
1025 INPUT "Ligne de depart (0-17):";p: LET p=p*8
1030 INPUT "Encre (0-7):";e: INK e: GO TO 100

```

Pour que le 2068 prononce MASTERMIND nous devons faire:

```
POKE 64901,60000-256*INT(60000/256)
POKE 64902,INT(60000/256)
POKE 64904,58500-256*INT(58500/256)
POKE 64905,INT(58500/256)
```

Si vous avez plusieurs mots il serait intéressant de définir ces fonctions au début de votre programme:

```
DEF FN l(x)=x-256*INT(x/256)
DEF FN h(x)=INT(x/256)
```

Ainsi notre exemple deviendrait:


```
POKE 64901,l(60000)
POKE 64902,h(60000)
POKE 64904,l(58500)
POKE 64905,h(58500)
```

Pour changer le délai ou le timbre de la voix, faites:

```
POKE 64950,délai désiré
```

Et enfin pour que le 2068 parle:

```
RANDOMIZE USR 64900
```



Liste: HEX-LOADER

1 REM

HEX - L O A D E R

```
5 CLEAR 31999: LET a=10: LET b=11: LET c=12: LET d=13: LET e=
14: LET f=15
10 INPUT "Adresse de depart:":adresse
20 LET ligne=100
30 READ s$,somme: LET tot=0
40 LET byte=16* VAL s$(1)+ VAL s$(2): LET tot=tot+byte: POKE a
dresse,byte
50 LET s$=s$(3 TO ): LET adresse=adresse+1: IF s$ <> "" THEN
GO TO 40
60 IF somme=tot THEN PRINT "Ligne ";ligne:" OK": LET ligne=l
igne+1: GO TO 30
70 BEEP .1,1: PRINT "Erreur a la ligne ";ligne: STOP
99 REM Placer ici les DATAs de la routine appropriee...
```

DATA pour caracteres agrandis

```
99 REM DATA [CODE caractere depart-->]32200 longueur:300

100 DATA "1600A770060517031210FB855F05C178C65847C84228047AC6075
70B4A2B047AC60E577AC64057C97CE6FB1F1F1FE567CDC87DE17CE6078257C97
DE6FB1F1F1F",7344
101 DATA "E56FCDF07DE17DE607C9CB7F280906004F21DB0709180326006F2
929297C063067C9CD027E2FE607171717F6C721045BCB7E2802D64032437E1AC
B0712C92118",6270
102 DATA "5B22005B2A005B7E2322005BFEFFC8CD127E22025B3E0832055B2
A075B220E5B2AD25B7E32045B2322025B3E0832065B2A0E5B22105B3A095B4F2
A105B220C5B",3911
103 DATA "3ADA5B47052A0C5B7CFC003801AF673200FBCD2A7E210C5B34C11
0E83A105B320C5B21005B340020D72A105B3A0A5B656F22105B3A045B1732045
B3A065B3032",4680
104 DATA "065B20B22A0E5B3A095B6467220E5B3A055B3D32055B20883A085
B4F3A045B3473A075B8110FD32075B034C7E000000000000000000000000000
00000000000",3083
```

DATA pour parler

99 REM DATA CODE parle depart--> 64900 longueur:69

100 DATA "2183FD1187DCE5A7ED52234C45E10C3A485C0F0F0FE607D94FE52
610D9F37E0F0F0F0FD9",3967


101 DATA "5F16087BA4B1D3FEC0B152806060A10FE18F0D92B3E0010E10D2
0DED9E1D9FBC9000000",3826

DATA pour entendre

99 REM DATA CODE entendre depart-->65000 longueur:78

100 DATA "2183FD11F57EE5A7ED52234C45E1383A0CF3167FD91680D93E7FD
BFE1F302BCB6F20F5D95ADBFE0B",5385

101 DATA "17C217CB133806060A10FE18F07BD9772B7ADBFE1F300510E20D2
0DF237E3C2BFB2B444DFBC900000",3934



TRUC 2068...TRUC 2068...TRUC 2068...TRUC 2068...TRUC 2068...TRUC

POKE 23693,56 ->Met l'écran avec les attributs suivants: PAPER 7
& INK 0.

POKE 23755,10 ->Rend LIST inopérant.

POKE 23755,0 ->Remet LIST en opération.

POKE 23613,PEEK (23730)-5 ->Rend BREAK inopérant.

LISTE: La Voix du TS2068

```

10 IF NOT charge THEN PRINT AT 2,5;"*****"
***      * LA VOIX DU TS 2068 *      *      GAGNON-c84
*      * M.C. tire de:      *      * SUPERCHARGE YOUR
*      * SPECTRUM de      *      * David Webb,Melbourn
e *      * House,1983      *      * *****
***: LET charge=1: RETURN
32 DEF FN l(x)=x-256* INT (x/256)
33 DEF FN h(x)= INT (x/256)
70 LET ent= VAL "65000": LET parle= VAL "64900": LET ze= SIN
PI : LET un= SGN PI : LET deux=un+un: LET trois= INT PI :
80 LET car= VAL "4000": LET dep= VAL "64899": LET limite= VAL
"32501": LET input= VAL "1000": LET menu= VAL "2000"

100 REM -----VOCABULAIRE
110 LET o$="LE VOCABULAIRE": GO SUB car: LET p$="No. de mots ma
ximums": GO SUB input: LET m= VAL i$: IF m<un THEN GO TO 110
120 LET p$="Longueur max. des mots": GO SUB input: LET n= VAL
i$: IF n<ze THEN GO TO 120
130 DIM a(m,trois): DIM n$(m,n)
140 FOR c=un TO m
150 LET p$=("Donnez le mot #"+ STR$ c+ " :"): GO SUB input: LET
a$=i$
160 LET n$(c)=a$( TO ( LEN a$ AND LEN a$ <= n)+(n AND LEN a$>
n))
170 GO SUB VAL "370"
180 IF a(c,trois)=limite THEN PRINT "'*Plus de place": PAUSE 1
00: GO TO menu
190 NEXT c
200 PRINT "'*Vocabulaire complet": PAUSE VAL "100": GO TO menu

210 REM -----Vue du Vocabulaire
220 LET o$="MOTS DISPONIBLES": GO SUB car: PRINT "No. MOT": TAB
VAL "14": "DELAI": TAB VAL "21": "DEPART": TAB VAL "28": "FIN":
PRINT
230 FOR c=un TO m
240 PRINT c: TAB trois;n$(c): TAB VAL "15":a(c,un): TAB VAL "
21":a(c,deux): TAB VAL "27":a(c,trois)
250 GO SUB VAL "520": NEXT c
270 PRINT AT VAL "20",ze;"*Une touche pour le menu.": PAUSE z

```



```

e: GO TO menu
280 REM -----ENTENDRE
290 LET 0$="AUDITION": GO SUB CAR: PRINT "Quel mot des ";m;" mots": LET p$="voulez-vous entendre": GO SUB input: LET c= VAL i$
300 IF c>m OR c<un THEN GO TO VAL "290"
310 PRINT ""Le mot: ";n$(c)"Le Delai:";a(c,un): GO SUB VAL "
520": PAUSE VAL "50": GO TO menu
320 REM -----CHANGEMENT
330 PRINT "Quel mots des ";m;" mots": LET p$="voulez-vous changer": GO SUB input: LET c= VAL i$
340 IF c>m OR c<un THEN GO TO VAL "330"
350 LET dep=a(c,deux): LET limite=a(c,trois)
360 GO SUB VAL "370": PAUSE VAL "50": GO TO menu

370 REM -----C.M. entendre
380 POKE ent+un, FN l(dep)
390 POKE ent+deux, FN h(dep)
400 POKE ent+ VAL "4", FN l(limite)
410 POKE ent+ VAL "5", FN h(limite)
420 LET p$="Delai (1-255)": GO SUB input: LET d= VAL i$
430 POKE ent+ VAL "4B",d
440 LET a(c,un)=d: LET a(c,deux)=dep
450 PRINT ""*Preparez votre cassette et faites ENTER pour la lecture": PAUSE ze
460 PRINT ""* O.K.""*Faites ESPACE ou attendez"" la fin de l'enregistrement."
470 BEEP VAL ".01", VAL "40": LET a(c,trois)=(USR ent)+un: BEP VAL ".1", VAL "40"
480 IF a(c,trois)>dep THEN PRINT "*Aucun son detecte...": PAUSE VAL "100": LET c=c-un: RETURN
490 GO SUB VAL "520": LET dep=a(c,trois)-un: RETURN

520 REM -----C.M. parle
530 POKE parle+un, FN l(a(c,deux))
540 POKE parle+deux, FN h(a(c,deux))
550 POKE parle+ VAL "4", FN l(a(c,trois))
570 POKE parle+ VAL "5", FN h(a(c,trois))
580 POKE parle+ VAL "50",a(c,un): LET a=USR parle: RETURN
610 CLEAR VAL "32199": PAPER 0: INK 7: BORDER 0: BRIGHT 0: CLS
: LET charge=0: GO SUB 10: PRINT AT 14,0;"UN INSTANT S.V.P.":
LOAD "" CODE : LOAD "" CODE : PRINT "UNE TOUCHE ": PAUSE 0: CLS

```

```

: GO TO 11
620 REM -----SAUVE
630 LET o$="SAUVE LES MOTS": GO SUB CAR: LET p$="Sauve du mot #
": GO SUB input: LET c1= VAL i$: LET p$="au mot #": GO SUB input
: LET c2= VAL i$
640 IF c1>m OR c2>m OR c1<ze OR c2<ze OR c1>c2 THEN GO TO VAL
"630"
650 LET st=a(c2,trois): LET le=a(c1,deux)-st+un
660 IF le<ze THEN PRINT "*LONGUEUR NEGATIVE": GO TO VAL "630"
670 PRINT "*PRENEZ NOTE"" ADRESSE de DEPART:":st" LONGUEUR:
":le
680 LET p$="Le Nom: ": GO SUB input: LET a$=i$
690 IF a$="" OR LEN a$>10 THEN GO TO VAL "680"
700 SAVE a$ CODE st,le
710 LET p$="Verification (o/n)": GO SUB input: IF i$(un)="o" O
R i$(un)="0" THEN VERIFY "a$" CODE
720 PRINT " O.K.": PAUSE VAL "70": GO TO menu


999 REM ----- INPUT
1000 INPUT ""
1001 IF INKEY$ <> "" THEN GO TO VAL "1000"
1005 BEEP VAL ".0001", VAL "60": PRINT p$;"_": LET i$=""
1015 FOR n=un TO VAL "10"
1020 IF INKEY$="" THEN GO TO VAL "1020"
1025 LET e$= INKEY$: IF CODE e$=13 THEN PRINT CHR$ VAL "8";
" ": PRINT : RETURN
1030 LET i$=i$+e$
1035 IF INKEY$ <> "" THEN GO TO VAL "1035"
1040 PRINT CHR$ VAL "8";i$(n);"_":
1045 NEXT n: PRINT : RETURN
2000 REM -----MENU
2010 LET o$="MENU": GO SUB car: PRINT AT VAL "21", VAL "21";"G
AGNON-o85": AT trois,ze;
2015 LET t$=" ": PRINT ""
2020 PRINT t$;"1. Refait le Vocabulaire"
2030 PRINT t$;"2. Vue du Vocabulaire"
2040 PRINT t$;"3. Entendre un mot"
2050 PRINT t$;"4. Change un mot"
2060 PRINT t$;"5. Sauve les mots"
2070 PRINT t$;"6. Change le delai(timbre)"
2071 PRINT t$;"7. Stop"

```

```

2075 PRINT ""Votre Choix:";
2078 IF INKEY$="" THEN GO TO VAL "2078"
2079 LET e$= INKEY$ : IF e$<"1" OR e$>"7" THEN GO TO VAL "2010"
"
2080 PRINT e$; AT 4+ VAL e$,trois; FLASH un;">": BEEP VAL ".01"
, VAL "40": PAUSE VAL "50"
2081 IF e$="1" THEN RUN VAL "31"
2082 CLS : GO TO ((210*(e$="2"))+(280*(e$="3"))+(320*(e$="4"))+(
620*(e$="5"))+(3000*(e$="6"))+(2084*(e$="7")))
2084 STOP
3000 REM _____delai
3005 LET o$="CORRECTION-DELAJ": GO SUB car
3010 PRINT "Quel des ";m;" mots": LET p$="voulez-vous modifie le
delai:": GO SUB input: LET w= VAL i$
3015 PRINT "Le mot est ";n$(w)""Le Delai au mot #";w;" est ";a(
w,un)'
3020 LET p$="Le nouveau delai (1-255):": GO SUB input: LET k= VA
L i$
3030 LET a(w,un)=k: LET c=w: GO SUB VAL "530"
3040 LET p$="Est-ce correct(o/n):": GO SUB input: IF i$(un)="n"
OR i$(un)="N" THEN GO TO VAL "3020"
3050 LET p$="Un autre mot(o/n):": GO SUB input: IF i$(un)="o" OR
i$(un)="O" THEN GO TO VAL "3010"
3060 GO TO menu
4000 REM _____Gros car.
4010 INK VAL "5": CLS : RESTORE : FOR i=un TO VAL "5": READ a$
: POKE 23302+i, VAL a$: NEXT i
4020 FOR i=un TO LEN o$: POKE 23319+i, CODE o$(i): NEXT i: POKE
23319+i,255: POKE 32314,40
4025 RANDOMIZE USR VAL "32326": PRINT AT trois,zei: INK VAL
"7": RETURN
4030 DATA "(256- LEN o$*16)/2","10","1","2","8"
5000 STOP
9999 SAVE "voix" LINE VAL "600": SAVE "ent&par" CODE 64900,180:
SAVE "car" CODE 32200,300: VERIFY "voix": VERIFY "ent&par" CODE
: VERIFY "car" CODE

```



Programme: TITRE 3D TS2068

Ce programme vous permet de créer des pages en 3D.

```

1 REM   ____ TITRE 3D ____
5 BORDER 0: PAPER 0: INK 7: CLS : INPUT "Le Message:"; LINE k
$: IF LEN k$>25 THEN LET k$=k$( TO 25)
20 FOR a=1 TO LEN k$: IF k$(a)=" " THEN NEXT a
30 LET si=a: LET x=a-1: LET y=0: LET l$=k$(a)
65 GO SUB 140: NEXT a: STOP
140 IF x>31 OR x<0 OR y<0 OR si+y>24 OR si<1 OR l$>" " OR l$<"
" THEN PRINT AT 0,31
150 LET cs=( PEEK 23606)+(( PEEK 23607)*256): LET c= CODE l$: L
ET cp=cs+(c*8)
180 GO SUB 300: LET co=0
200 FOR s=1 TO si
210 FOR n=0 TO 7: LET pos=16384+((y-f)*32)+y1+x+(n*256): IF co=
si THEN LET cp=cp+1: LET co=0
240 LET co=co+1: POKE pos, PEEK cp: NEXT n: LET y=y+1
280 IF y=8 OR y=16 THEN GO SUB 300
290 NEXT s: RETURN
300 LET y1= INT (y/8): LET f=y1*8: LET y1=y1*2048: RETURN
9999 SAVE "TITRE3D" LINE 1: VERIFY ""

```

PROGRAMME: SURFACE PLANE TS2068

```

1 REM SURFACE PLANE
10 FOR X=-1 TO 1.2 STEP 1/5: FOR Y=-1 TO 1 STEP 1/75
20 GO SUB 1000: NEXT Y: NEXT X
30 FOR Y=-1 TO 1.2 STEP 1/5: FOR X=-1 TO 1 STEP 1/75
40 GO SUB 1000: NEXT X: NEXT Y: STOP
1000 LET Z=Y*X*X: LET P=(X+Y*.7)*50: LET Q=(Z+Y*.7)*50
1010 PLOT 125+P,85-Q: RETURN
4999 REM Pour obtenir des figures
5000 REM differentes, changer la la valeur de la variable z (lign
e 1000)
5005 REM Voici quelques valeurs interessantes que peut prendre
z:
5010 LET z=y* SQR ABS (x*y)
5020 LET z=x*x*x*x
5030 LET z= SQR (y*y*x*x*x)

```